Differential operators in Spherical coordinate with the use of Mathematica

Masatsugu Sei Suzuki
Department of Physics, SUNY at Binghamton
(Date: February 07, 2021)

The differential operator is one of the most important programs in Mathematica. The use of such techniques makes one so easy to solve the Schrodinger equation, and treat the commutation relations of angular momentum and linear momentum. Here we discuss the differential operators in the spherical coordinates with the use of Mathematica.



Fig. Spherical coordinates.

1. Vector analysis in spherical coordinates

$$\mathbf{r} = \mathbf{e}_x r\sin\theta\cos\phi + \mathbf{e}_y r\sin\theta\sin\phi + \mathbf{e}_z r\cos\theta$$

The spherical polar unit vectors in terms of cartesian unit vectors,

$$\mathbf{e}_r = \frac{\partial \mathbf{r}}{\partial r} = \mathbf{r} = \mathbf{e}_x \sin\theta\cos\phi + \mathbf{e}_y \sin\theta\sin\phi + \mathbf{e}_z \cos\theta,$$

$$\mathbf{e}_\theta = \frac{1}{r}\frac{\partial \mathbf{r}}{\partial \theta} = \mathbf{e}_x \cos\theta\cos\phi + \mathbf{e}_y \cos\theta\sin\phi - \mathbf{e}_z \sin\theta,$$

$$\mathbf{e}_\phi = \frac{1}{r\sin\theta}\frac{\partial \mathbf{r}}{\partial \phi} = -\mathbf{e}_x \sin\phi + \mathbf{e}_y \cos\phi,$$

$$\begin{pmatrix} \mathbf{e}_r \\ \mathbf{e}_\theta \\ \mathbf{e}_\phi \end{pmatrix} = \begin{pmatrix} \sin\theta\cos\phi & \sin\theta\sin\phi & \cos\theta \\ \cos\theta\cos\varphi & \cos\theta\sin\phi & -\sin\theta \\ -\sin\phi & \cos\phi & 0 \end{pmatrix} \begin{pmatrix} \mathbf{e}_x \\ \mathbf{e}_y \\ \mathbf{e}_z \end{pmatrix}.$$

The Cartesian unit vectors in terms of spherical polar unit vectors,

$$\begin{pmatrix} \mathbf{e}_x \\ \mathbf{e}_y \\ \mathbf{e}_z \end{pmatrix} = \begin{pmatrix} \sin\theta\cos\phi & \cos\theta\cos\phi & -\sin\phi \\ \sin\theta\sin\phi & \cos\theta\sin\phi & \cos\phi \\ \cos\theta & -\sin\theta & 0 \end{pmatrix} \begin{pmatrix} \mathbf{e}_r \\ \mathbf{e}_\theta \\ \mathbf{e}_\phi \end{pmatrix},$$

$$\mathbf{e}_x = \mathbf{e}_r \sin\theta\cos\phi + \mathbf{e}_\theta \cos\theta\cos\phi - \mathbf{e}_\phi \sin\phi,$$

$$\mathbf{e}_y = \mathbf{e}_r \sin\theta\sin\phi + \mathbf{e}_\theta \cos\theta\sin\phi + \mathbf{e}_\phi \cos\phi,$$

$$\mathbf{e}_z = \mathbf{e}_r \cos\theta - \mathbf{e}_\theta \sin\theta.$$

$$\mathbf{r} = \mathbf{e}_r r = x\mathbf{e}_x + y\mathbf{e}_y + z\mathbf{e}_z.$$

$$r = |\mathbf{r}| = \sqrt{x^2 + y^2 + z^2}.$$

$$d\mathbf{r} = \mathbf{e}_r dr + \mathbf{e}_\theta r d\theta + \mathbf{e}_\phi r\sin\theta d\phi.$$

The radial linear momentum

$$p_r = \frac{1}{2}(\frac{\mathbf{r}}{r}\cdot\mathbf{p} + \mathbf{p}\cdot\frac{\mathbf{r}}{r}) \qquad\qquad p_r = \frac{\hbar}{ir}\frac{\partial}{\partial r}(r)$$

((**Mathematica**))

Here we use the following notations in the Mathematica program for the spherical coordinates.

$$ux \to \mathbf{e}_x = \mathbf{e}_r \sin\theta\cos\phi + \mathbf{e}_\theta \cos\theta\cos\phi - \mathbf{e}_\phi \sin\phi$$

$$u_y \to \mathbf{e}_y = \mathbf{e}_r \sin\theta\sin\phi + \mathbf{e}_\theta \cos\theta\sin\phi + \mathbf{e}_\phi \cos\phi$$

$$uz \to \mathbf{e}_z = \mathbf{e}_r \cos\theta - \mathbf{e}_\theta \sin\theta$$

$$u_r \to \mathbf{e}_r$$

$$x \to \mathbf{e}_x \cdot r\mathbf{e}_r$$

$$y \to \mathbf{e}_y \cdot r\mathbf{e}_r$$

$$z \to \mathbf{e}_z \cdot r\mathbf{e}_r$$

$$\text{Lap} \to \nabla^2$$

$$\text{Gra} \to grad,\ \ \nabla$$

$$\text{Diva} \to div$$

$$\text{Curla} \to curl \quad \text{or} \quad rot$$

$$L \to \mathbf{L} = \mathbf{r}\times\mathbf{p}$$

$$L_x \to \mathbf{e}_x \cdot \mathbf{L}$$

$$L_y \to \mathbf{e}_y \cdot \mathbf{L}$$

$$L_z \to \mathbf{e}_z \cdot \mathbf{L}$$

$$LP \to L_x + iL_y$$

$$LM \to L_x - iL_y$$

$$Leq \to \mathbf{L}^2 = L_x{}^2 + L_y{}^2 + L_z{}^2$$

$$p \to \mathbf{p} = \frac{\hbar}{i}\nabla$$

$$p_x \to \mathbf{e}_x \cdot \mathbf{p} = \frac{\hbar}{i}\mathbf{e}_x \cdot \nabla$$

$$p_y \to \mathbf{e}_y \cdot \mathbf{p} = \frac{\hbar}{i}\mathbf{e}_y \cdot \nabla\ .$$

$$p_z \to \mathbf{e}_z \cdot \mathbf{p} = \frac{\hbar}{i}\mathbf{e}_z \cdot \nabla\ .$$

$Peq \to \mathbf{p}^2 = p_x{}^2 + p_y{}^2 + p_z{}^2$.

$prq \to p_r$ (radial linear momentum).

$pq \to \dfrac{\hbar}{ir}\dfrac{\partial}{\partial r} r$ (radial linear momentum, simplified form).

$\mathbf{r} \to r\mathbf{e}_r$

$x \to \mathbf{e}_x \cdot r\mathbf{e}_r = r\sin\theta\cos\phi$.

$y \to \mathbf{e}_y \cdot r\mathbf{e}_r = r\sin\theta\sin\phi$.

$z \to r\mathbf{e}_r \cdot \mathbf{e}_z = r\cos\theta$.

```
Clear["Global`"];
ux = {Sin[θ] Cos[ϕ], Cos[θ] Cos[ϕ], -Sin[ϕ]};
uy = {Sin[θ] Sin[ϕ], Cos[θ] Sin[ϕ], Cos[ϕ]};
uz = {Cos[θ], -Sin[θ], 0};
ur = {1, 0, 0};
x = r ur.ux;
y = r ur.uy;
z = r ur.uz;
 Lap := Laplacian[#, { r, θ, ϕ}, "Spherical"] &;
Gra := Grad[#, { r, θ, ϕ}, "Spherical"] &;
Diva := Div[#, {r, θ, ϕ}, "Spherical"] &;
Curla := Curl[#, {r, θ, ϕ}, "Spherical"] &;
L := (-i ħ ( Cross[(ur r), Gra[#]]) &) // Simplify;
Lx := (ux.L[#] &) // Simplify;
Ly := (uy.L[#] &) // Simplify;
Lz := (uz.L[#] &) // Simplify;
LP := (Lx[#] + i Ly[#]) & // Simplify;
LM = (Lx[#] - i Ly[#]) & // Simplify;
Leq := (Nest[Lx, #, 2] + Nest[Ly, #, 2] + Nest[Lz, #, 2]) &;
p := (-i ħ Gra[#]) &;
px := (ux.p[#]) &;
py := (uy.p[#]) &;
pz := (uz.p[#]) &;
Peq := (Nest[px, #, 2] + Nest[py, #, 2] + Nest[pz, #, 2]) &;
prq := ((-i ħ)/2 ur .Gra[#] + (-i ħ)/2 Diva[# ur]) &;
pq := ħ/(i r) D[r #, r] &;
```

2. Grad, Laplacian, Div, Curl

$$\nabla = \mathbf{e}_r \frac{\partial}{\partial r} + \mathbf{e}_\theta \frac{1}{r}\frac{\partial}{\partial \theta} + \mathbf{e}_\phi \frac{1}{r\sin\theta}\frac{\partial}{\partial \phi}$$

$$\nabla^2 = \frac{1}{r^2}(2r\frac{\partial}{\partial r} + r^2\frac{\partial^2}{\partial r^2}) + \frac{1}{r^2\sin^2\theta}\frac{\partial^2}{\partial \phi^2}$$
$$+\frac{1}{r^2}(\cot\theta\frac{\partial}{\partial\theta} + \frac{\partial^2}{\partial\theta^2})$$
$$= \frac{1}{r^2}\frac{\partial}{\partial r}(r^2\frac{\partial}{\partial r}) + \frac{1}{r^2\sin\theta}\frac{\partial}{\partial\theta}(\sin\theta\frac{\partial}{\partial\theta}) + \frac{1}{r^2\sin^2\theta}\frac{\partial^2}{\partial\phi^2}$$

$$\nabla\cdot\mathbf{V} = \frac{1}{r}(2V_r + r\frac{\partial V_r}{\partial r}) + \frac{1}{r}(\cot\theta V_\theta + \frac{\partial V_\theta}{\partial\theta}) + \frac{1}{r\sin\theta}\frac{\partial V_\phi}{\partial\phi}$$
$$= \frac{1}{r^2}\frac{\partial}{\partial r}(r^2 V_r) + \frac{1}{r\sin\theta}\frac{\partial}{\partial\theta}(\sin\theta V_\theta) + \frac{1}{r\sin\theta}\frac{\partial V_\phi}{\partial\phi}$$

$$\nabla^2\psi = \frac{1}{r^2}\frac{\partial}{\partial r}(r^2\frac{\partial\psi}{\partial r}) + \frac{1}{r^2\sin\theta}\frac{\partial}{\partial\theta}(\sin\theta\frac{\partial\psi}{\partial\theta}) + \frac{1}{r^2\sin^2\theta}\frac{\partial^2\psi}{\partial\phi^2}$$

$$\nabla\times\mathbf{V} = \frac{1}{r^2\sin\theta}\begin{vmatrix} \mathbf{e}_r & r\mathbf{e}_\theta & r\sin\theta\mathbf{e}_\phi \\ \frac{\partial}{\partial r} & \frac{\partial}{\partial\theta} & \frac{\partial}{\partial\phi} \\ V_r & rV_\theta & r\sin\theta V_\phi \end{vmatrix}$$

((**Mathematica**))

```
eq1 = Gra[ψ[r, θ, ϕ]] // Simplify
```

$$\left\{\psi^{(1,0,0)}[r,\theta,\phi],\ \frac{\psi^{(0,1,0)}[r,\theta,\phi]}{r},\ \frac{\mathrm{Csc}[\theta]\,\psi^{(0,0,1)}[r,\theta,\phi]}{r}\right\}$$

```
eq2 = Lap[ψ[r, θ, ϕ]] // Simplify
```

$$\frac{1}{r^2}\left(\mathrm{Csc}[\theta]^2\,\psi^{(0,0,2)}[r,\theta,\phi]+\mathrm{Cot}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi]+\psi^{(0,2,0)}[r,\theta,\phi]+2\,r\,\psi^{(1,0,0)}[r,\theta,\phi]+r^2\,\psi^{(2,0,0)}[r,\theta,\phi]\right)$$

```
eq3 = Diva[{Vr[r, θ, ϕ], Vθ[r, θ, ϕ], Vϕ[r, θ, ϕ] }] //
  Simplify
```

$$\frac{1}{r}\left(2\,\mathrm{Vr}[r,\theta,\phi]+\mathrm{Cot}[\theta]\,\mathrm{V\theta}[r,\theta,\phi]+\mathrm{Csc}[\theta]\,\mathrm{V\phi}^{(0,0,1)}[r,\theta,\phi]+\mathrm{V\theta}^{(0,1,0)}[r,\theta,\phi]+r\,\mathrm{Vr}^{(1,0,0)}[r,\theta,\phi]\right)$$

```
eq4 = Curla[{Vr[r, θ, ϕ], Vθ[r, θ, ϕ], Vϕ[r, θ, ϕ] }] //
  Simplify
```

$$\left\{\frac{1}{r}\left(\mathrm{Cot}[\theta]\,\mathrm{V\phi}[r,\theta,\phi]-\mathrm{Csc}[\theta]\,\mathrm{V\theta}^{(0,0,1)}[r,\theta,\phi]+\mathrm{V\phi}^{(0,1,0)}[r,\theta,\phi]\right),\ -\frac{1}{r}\left(\mathrm{V\phi}[r,\theta,\phi]-\mathrm{Csc}[\theta]\,\mathrm{Vr}^{(0,0,1)}[r,\theta,\phi]+r\,\mathrm{V\phi}^{(1,0,0)}[r,\theta,\phi]\right),\ \frac{\mathrm{V\theta}[r,\theta,\phi]-\mathrm{Vr}^{(0,1,0)}[r,\theta,\phi]+r\,\mathrm{V\theta}^{(1,0,0)}[r,\theta,\phi]}{r}\right\}$$

3. Angular momentum

In quantum mechanics, the angular momentum is defined as folloews.

$$\mathbf{L}=\mathbf{r}\times\mathbf{p}=\frac{\hbar}{i}\mathbf{r}\times\nabla=-i\hbar(\mathbf{e}_\phi\frac{\partial}{\partial\theta}-\mathbf{e}_\theta\frac{1}{\sin\theta}\frac{\partial}{\partial\phi})$$

$$L_x = \mathbf{e}_x \cdot \mathbf{L} = i\hbar(\sin\phi \frac{\partial}{\partial\theta} + \cot\theta\cos\phi \frac{\partial}{\partial\phi})$$

$$L_y = \mathbf{e}_y \cdot \mathbf{L} = i\hbar(-\cos\phi \frac{\partial}{\partial\theta} + \cot\theta\sin\phi \frac{\partial}{\partial\phi})$$

$$L_z = \mathbf{e}_z \cdot \mathbf{L} = -i\hbar \frac{\partial}{\partial\phi}$$

The Ladder operator:

$$L_+ = L_x + iL_y = \hbar e^{i\phi}(\frac{\partial}{\partial\theta} + i\cot\theta \frac{\partial}{\partial\phi})$$

$$L_- = L_x - iL_y = -\hbar e^{-i\phi}(\frac{\partial}{\partial\theta} - i\cot\theta \frac{\partial}{\partial\phi})$$

$$\begin{aligned}
\mathbf{L}^2 &= L_x{}^2 + L_y{}^2 + L_z{}^2 \\
&= -\hbar^2[\csc^2(\theta)\frac{\partial^2}{\partial\phi^2} + \cot\theta\frac{\partial}{\partial\theta} + \frac{\partial^2}{\partial\theta^2}] \\
&= -\hbar^2[\frac{1}{\sin\theta}\frac{\partial}{\partial\theta}(\sin\theta\frac{\partial}{\partial\theta}) + \frac{1}{\sin^2(\theta)}\frac{\partial^2}{\partial\phi^2}]
\end{aligned}$$

((**Mathematica**))

```
eq0 = L[ψ[r, θ, ϕ]] // Simplify
{0, i ℏ Csc[θ] ψ^(0,0,1)[r, θ, ϕ], -i ℏ ψ^(0,1,0)[r, θ, ϕ]}

eq1 = Lx[ψ[r, θ, ϕ]] // Simplify
i ℏ (Cos[ϕ] Cot[θ] ψ^(0,0,1)[r, θ, ϕ] +
   Sin[ϕ] ψ^(0,1,0)[r, θ, ϕ])

eq2 = Ly[ψ[r, θ, ϕ]] // Simplify
i ℏ (Cot[θ] Sin[ϕ] ψ^(0,0,1)[r, θ, ϕ] -
   Cos[ϕ] ψ^(0,1,0)[r, θ, ϕ])

eq3 = Lz[ψ[r, θ, ϕ]] // Simplify
-i ℏ ψ^(0,0,1)[r, θ, ϕ]

eq4 = LP[ψ[r, θ, ϕ]] // Simplify
ℏ (Cos[ϕ] + i Sin[ϕ])
 (i Cot[θ] ψ^(0,0,1)[r, θ, ϕ] + ψ^(0,1,0)[r, θ, ϕ])

eq5 = LM[ψ[r, θ, ϕ]] // Simplify
ℏ (i Cos[ϕ] + Sin[ϕ])
 (Cot[θ] ψ^(0,0,1)[r, θ, ϕ] + i ψ^(0,1,0)[r, θ, ϕ])

eq6 = Leq[ψ[r, θ, ϕ]] // Simplify
-ℏ^2 (Csc[θ]^2 ψ^(0,0,2)[r, θ, ϕ] +
   Cot[θ] ψ^(0,1,0)[r, θ, ϕ] + ψ^(0,2,0)[r, θ, ϕ])
```

4. Radial linear momentum

The radial linear momentum in the quantum mechanics is defined by

$$p_r = \frac{1}{2}(\frac{\mathbf{r}}{r}\cdot\mathbf{p} + \mathbf{p}\cdot\frac{\mathbf{r}}{r}) = \frac{\hbar}{2i}(\frac{\mathbf{r}}{r}\cdot\nabla + \nabla\cdot\frac{\mathbf{r}}{r})$$

or

$$p_r = \frac{\hbar}{ir}\frac{\partial}{\partial r} r = \frac{\hbar}{i}(\frac{1}{r} + \frac{\partial}{\partial r}).$$

Commutation relation

$$[p_r, r] = \frac{\hbar}{i}.$$

((**Mathematica**))

Radial linear momentum

```
prq[ψ[r, θ, ϕ]] // Simplify
```

$$-\frac{i\hbar\left(\psi[r,\theta,\phi]+r\,\psi^{(1,0,0)}[r,\theta,\phi]\right)}{r}$$

Commutation relation: prq r - r prq=-iℏ

```
prq[r ψ[r, θ, ϕ]] − r prq[ψ[r, θ, ϕ]] // Simplify
```

$-i\hbar\,\psi[r,\theta,\phi]$

5. Relation between *p* and *L*

We show that $\mathbf{p}^2$ can be expressed by

$$\mathbf{p}^2 = p_r{}^2 + \frac{1}{r^2}\mathbf{L}^2$$

where

$$\begin{aligned}\mathbf{p}^2 &= -\hbar^2\nabla^2 \\ &= -\hbar^2[\frac{1}{r^2}\frac{\partial}{\partial r}(r^2\frac{\partial}{\partial r}) + \frac{1}{r^2\sin\theta}\frac{\partial}{\partial\theta}(\sin\theta\frac{\partial}{\partial\theta}) + \frac{1}{r^2\sin^2\theta}\frac{\partial^2}{\partial\phi^2}]\end{aligned}$$

$$\mathbf{L}^2 = L_x{}^2 + L_y{}^2 + L_z{}^2 = -\hbar^2[\frac{1}{\sin\theta}\frac{\partial}{\partial\theta}(\sin\theta\frac{\partial}{\partial\theta}) + \frac{1}{\sin^2(\theta)}\frac{\partial^2}{\partial\phi^2}]$$

$$p_r{}^2 = \frac{\hbar}{ir}\frac{\partial}{\partial r}(r\frac{\hbar}{ir}\frac{\partial}{\partial r}r) = -\frac{\hbar^2}{r}\frac{\partial^2}{\partial r^2}r = -\frac{\hbar^2}{r^2}\frac{\partial}{\partial r}(r^2\frac{\partial}{\partial r})$$

$$\begin{aligned}\mathbf{p}^2 &= p_r{}^2 + \frac{1}{r^2}\mathbf{L}^2 \\ &= -\hbar^2[\frac{1}{r}\frac{\partial^2}{\partial r^2}r + \frac{1}{r^2\sin\theta}\frac{\partial}{\partial\theta}(\sin\theta\frac{\partial}{\partial\theta}) + \frac{1}{r^2\sin^2(\theta)}\frac{\partial^2}{\partial\phi^2}] \\ &= -\hbar^2[\frac{1}{r^2}\frac{\partial}{\partial r}(r^2\frac{\partial}{\partial r}) + \frac{1}{r^2\sin\theta}\frac{\partial}{\partial\theta}(\sin\theta\frac{\partial}{\partial\theta}) + \frac{1}{r^2\sin^2(\theta)}\frac{\partial^2}{\partial\phi^2}]\end{aligned}$$

((**Mathematica**))

```
eq1 = Peq[ψ[r, θ, ϕ]] // Simplify
```

$$-\frac{1}{r^2}\hbar^2 \left(\text{Csc}[\theta]^2\,\psi^{(0,0,2)}[r,\theta,\phi] + \text{Cot}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi] + \psi^{(0,2,0)}[r,\theta,\phi] + 2\,r\,\psi^{(1,0,0)}[r,\theta,\phi] + r^2\,\psi^{(2,0,0)}[r,\theta,\phi]\right)$$

```
eq2 = Nest[prq, ψ[r, θ, ϕ], 2] // Simplify
```

$$-\frac{\hbar^2\left(2\,\psi^{(1,0,0)}[r,\theta,\phi] + r\,\psi^{(2,0,0)}[r,\theta,\phi]\right)}{r}$$

$$\text{eq3} = \frac{1}{r^2}\,\text{Leq}[\psi[r,\theta,\phi]]\ //\ \text{FullSimplify}$$

$$-\frac{1}{r^2}\hbar^2 \left(\text{Csc}[\theta]^2\,\psi^{(0,0,2)}[r,\theta,\phi] + \text{Cot}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi] + \psi^{(0,2,0)}[r,\theta,\phi]\right)$$

```
eq23 = eq2 + eq3 // Simplify
```

$$-\frac{1}{r^2}\hbar^2 \left(\text{Csc}[\theta]^2\,\psi^{(0,0,2)}[r,\theta,\phi] + \text{Cot}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi] + \psi^{(0,2,0)}[r,\theta,\phi] + 2\,r\,\psi^{(1,0,0)}[r,\theta,\phi] + r^2\,\psi^{(2,0,0)}[r,\theta,\phi]\right)$$

```
eq1 - eq23 // Simplify
```

0

6. Commutation relation

$$[L_i, x_j] = i\hbar \sum_k \varepsilon_{ijk} x_k$$

$$[L_x, L_y] = i\hbar L_z, [L_y, L_z] = i\hbar L_x, [L_z, L_x] = i\hbar L_y$$

$$[\mathbf{L}^2, L_x] = 0, \qquad [\mathbf{L}^2, L_y] = 0, \qquad [\mathbf{L}^2, L_z] = 0$$

$$[\mathbf{L}^2, L_\pm] = 0, \qquad [L_+, L_-] = 2\hbar \hat{L}_z,$$

$$[L_z, L_+] = \hbar L_+ \qquad [L_z, L_-] = -\hbar L_-$$

$$[L_+, L_-] = 2\hbar L_z$$

$$\mathbf{L}^2 = \frac{1}{2}(L_+ L_- + L_- L_+) + L_z^2$$

((**Levi-Civita symbol**))

ε_{ijk} is the Levi Civita symbol. $\varepsilon_{123} = \varepsilon_{231} = \varepsilon_{312} = 1$ (even permutation). $\varepsilon_{321} = \varepsilon_{132} = \varepsilon_{213} = -1$ (odd permutation). $\varepsilon_{ijk} == 0$ if $i = j$, or $j = k$, or $k = i$.

((**Mathematica**))

The commutation relations can be easily proved through Mathematica

```
Lx[y ψ[r, θ, ϕ]] - y Lx[ ψ[r, θ, ϕ]] -
  ⅈ ℏ z ψ[r, θ, ϕ] // FullSimplify

0

Ly[z ψ[r, θ, ϕ]] - z Ly[ ψ[r, θ, ϕ]] -
  ⅈ ℏ x ψ[r, θ, ϕ] // FullSimplify

0

Lz[x ψ[r, θ, ϕ]] - x Lz[ ψ[r, θ, ϕ]] -
  ⅈ ℏ y ψ[r, θ, ϕ] // FullSimplify

0

Lx[Ly [ψ[r, θ, ϕ]]] - Ly[Lx[ ψ[r, θ, ϕ]]] -
  ⅈ ℏ Lz [ψ[r, θ, ϕ]] // FullSimplify

0

Ly[Lz [ψ[r, θ, ϕ]]] - Lz[Ly[ ψ[r, θ, ϕ]]] -
  ⅈ ℏ Lx [ψ[r, θ, ϕ]] // FullSimplify

0

Lz[Lx [ψ[r, θ, ϕ]]] - Lx[Lz[ ψ[r, θ, ϕ]]] -
  ⅈ ℏ Ly [ψ[r, θ, ϕ]] // FullSimplify

0
```

```
Leq[Ly [ψ[r, θ, ϕ]]] - Ly[Leq[ ψ[r, θ, ϕ]]] //
 FullSimplify

0

LP[LM [ψ[r, θ, ϕ]]] - LM[LP[ ψ[r, θ, ϕ]]] -
  2 ℏ Lz[ ψ[r, θ, ϕ]] // FullSimplify

0

Leq[Lz [ψ[r, θ, ϕ]]] - Lz[Leq[ ψ[r, θ, ϕ]]] //
 FullSimplify

0

Leq[LP [ψ[r, θ, ϕ]]] - LP[Leq[ ψ[r, θ, ϕ]]] //
  FullSimplify // FullSimplify

0

Leq[LM [ψ[r, θ, ϕ]]] - LM[Leq[ ψ[r, θ, ϕ]]] //
  FullSimplify // FullSimplify

0

Lz[LP [ψ[r, θ, ϕ]]] - LP[Lz[ ψ[r, θ, ϕ]]] -
  ℏ LP[ ψ[r, θ, ϕ]] // FullSimplify

0

Lz[LM [ψ[r, θ, ϕ]]] - LM[Lz[ ψ[r, θ, ϕ]]] +
  ℏ LM[ ψ[r, θ, ϕ]] // FullSimplify

0
```

```
eq1 = (LP[LM [ψ[r, θ, ϕ]]] + LM[LP[ ψ[r, θ, ϕ]]]) /
    2 // FullSimplify;
eq2 = Lx[Lx [ψ[r, θ, ϕ]]] + Ly[Ly[ ψ[r, θ, ϕ]]] //
   FullSimplify;

(eq1 - eq2) // Simplify

0
```

7. Relations

$$[L^2, x_k] = (-i\hbar)\sum_{i,j} \varepsilon_{ijk}(L_i x_j + x_j L_i)$$

We show that

$$[\mathbf{L}^2, x] = (-i\hbar)(L_y z + zL_y - L_z y - yL_z)$$
$$= 2r\hbar^2(\cos\phi\sin\theta + \sin\phi\csc\theta\frac{\partial}{\partial\phi} - \cos\phi\cos\theta\frac{\partial}{\partial\theta})$$

$$[\mathbf{L}^2, y] = (-i\hbar)(L_z x + xL_z - L_x z - zL_x)$$
$$= 2r\hbar^2(\sin\phi\sin\theta - \cos\phi\csc\theta\frac{\partial}{\partial\phi} - \sin\phi\cos\theta\frac{\partial}{\partial\theta})$$

$$[\mathbf{L}^2, z] = (-i\hbar)(L_x y + yL_x - L_y x - xL_y)$$
$$= 2r\hbar^2(\cos\theta + \sin\theta\frac{\partial}{\partial\theta})$$

through Mathematica.

((**Mathematica**))

```
eq1 = Leq[x ψ[r, θ, ϕ]] - x Leq[ ψ[r, θ, ϕ]] //
  FullSimplify
```

$$2\, r\, \hbar^2 \left(\text{Cos}[\phi]\, \text{Sin}[\theta]\, \psi[r, \theta, \phi] + \text{Csc}[\theta]\, \text{Sin}[\phi]\, \psi^{(0,0,1)}[r, \theta, \phi] - \text{Cos}[\theta]\, \text{Cos}[\phi]\, \psi^{(0,1,0)}[r, \theta, \phi]\right)$$

```
eq11 =
 (-ⅈ ℏ) (Ly[z ψ[r, θ, ϕ]] + z Ly[ ψ[r, θ, ϕ]] -
    Lz[y ψ[r, θ, ϕ]] - y Lz[ ψ[r, θ, ϕ]]) // Simplify
```

$$2\, r\, \hbar^2 \left(\text{Cos}[\phi]\, \text{Sin}[\theta]\, \psi[r, \theta, \phi] + \text{Csc}[\theta]\, \text{Sin}[\phi]\, \psi^{(0,0,1)}[r, \theta, \phi] - \text{Cos}[\theta]\, \text{Cos}[\phi]\, \psi^{(0,1,0)}[r, \theta, \phi]\right)$$

```
eq1 - eq11 // Simplify
```

0

```
eq2 = Leq[y ψ[r, θ, ϕ]] - y Leq[ ψ[r, θ, ϕ]] // Simplify
```

$$r\hbar^2 \left(2\,\text{Sin}[\theta]\,\text{Sin}[\phi]\,\psi[r,\theta,\phi] - 2\left(\text{Cos}[\phi]\,\text{Csc}[\theta]\,\psi^{(0,0,1)}[r,\theta,\phi] + \text{Cos}[\theta]\,\text{Sin}[\phi]\,\psi^{(0,1,0)}[r,\theta,\phi]\right)\right)$$

```
eq21 =
 (-ⅈ ℏ) (Lz[x ψ[r, θ, ϕ]] + x Lz[ ψ[r, θ, ϕ]] -
     Lx[z ψ[r, θ, ϕ]] - z Lx[ ψ[r, θ, ϕ]]) // Simplify
```

$$-2r\hbar^2 \left(-\text{Sin}[\theta]\,\text{Sin}[\phi]\,\psi[r,\theta,\phi] + \text{Cos}[\phi]\,\text{Csc}[\theta]\,\psi^{(0,0,1)}[r,\theta,\phi] + \text{Cos}[\theta]\,\text{Sin}[\phi]\,\psi^{(0,1,0)}[r,\theta,\phi]\right)$$

```
eq2 - eq21 // Simplify
```

0

```
eq3 = Leq[z ψ[r, θ, ϕ]] - z Leq[ ψ[r, θ, ϕ]] //
  FullSimplify
```

$$2r\hbar^2 \left(\text{Cos}[\theta]\,\psi[r,\theta,\phi] + \text{Sin}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi]\right)$$

```
eq31 =
 (-ⅈ ℏ) (Lx[y ψ[r, θ, ϕ]] + y Lx[ ψ[r, θ, ϕ]] -
     Ly[ x ψ[r, θ, ϕ]] - x Ly[ ψ[r, θ, ϕ]]) // Simplify
```

$$2r\hbar^2 \left(\text{Cos}[\theta]\,\psi[r,\theta,\phi] + \text{Sin}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi]\right)$$

```
eq3 - eq31 // Simplify
```

0

8. Formula

We show that

$$[\mathbf{L}^2,[\mathbf{L}^2,x]]=2\hbar^2(x\mathbf{L}^2+\mathbf{L}^2x)$$

((**Proof**))
((**Mathematica**))

```
eq1 =
 Leq[Leq[x ψ[r, θ, ϕ]]] - Leq[x Leq[ ψ[r, θ, ϕ]]] -
   Leq[x Leq[ψ[r, θ, ϕ]]] + x Leq[ Leq[ψ[r, θ, ϕ]]] //
  FullSimplify
```

$$2\,r\,\hbar^4\left(2\,\mathrm{Csc}[\theta]\,\mathrm{Sin}[\phi]\,\psi^{(0,0,1)}[r,\theta,\phi]-2\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta]\left(-\psi[r,\theta,\phi]+\mathrm{Csc}[\theta]^2\,\psi^{(0,0,2)}[r,\theta,\phi]+2\,\mathrm{Cot}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi]+\psi^{(0,2,0)}[r,\theta,\phi]\right)\right)$$

```
eq2 = 2 ħ^2 (x Leq[ψ[r, θ, ϕ]] + Leq[ x ψ[r, θ, ϕ]]) //
  FullSimplify
```

$$2\,r\,\hbar^4\left(2\,\mathrm{Csc}[\theta]\,\mathrm{Sin}[\phi]\,\psi^{(0,0,1)}[r,\theta,\phi]-2\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta]\left(-\psi[r,\theta,\phi]+\mathrm{Csc}[\theta]^2\,\psi^{(0,0,2)}[r,\theta,\phi]+2\,\mathrm{Cot}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi]+\psi^{(0,2,0)}[r,\theta,\phi]\right)\right)$$

```
eq1 - eq2
```

0

9. Commutation relations for the radial momentum

$$[r,p_r]=i\hbar$$

$$[r,p_r^{\ 2}]=i2\hbar p_r$$

$$[r,p_r^{\ 3}]=i3\hbar p_r^{\ 2}$$

$$[r,p_r^{\ n}]=in\hbar p_r^{\ n-1}$$

$$[p_r,r]=-i\hbar$$

$$[p_r,r^2]=-i2\hbar r\ .$$

$$[p_r, r^3] = -i3\hbar r^2 .$$

$$[p_r, r^n] = -in\hbar r^{n-1} .$$

$$[p_r, \frac{1}{r}] = \frac{i\hbar}{r^2} .$$

$$[p_r, \frac{1}{r^2}] = \frac{2i\hbar}{r^3} .$$

$$[p_r{}^2, \frac{1}{r^2}] = \frac{-2\hbar^2}{r^4}(1 - 2r\frac{\partial}{\partial r}) .$$

$$\begin{aligned}[p_r{}^2, \frac{1}{r}] &= -[\frac{1}{r}, p_r{}^2] \\ &= 2[p_r, \frac{1}{r}]p_r \\ &= \frac{2i\hbar}{r^2} p_r \\ &= \frac{2\hbar^2}{r^2}\frac{\partial}{\partial r}\end{aligned}$$

((**Mathematica**))

Commutation relations of quantum mechanical radial angular momentum and power of r

```
Clear["Global`*"]; Pr = (1/r) (ℏ/ⅈ) D[r #, r] &;
```

```
Pr[Pr[ψ[r, θ, ϕ]]] // Simplify
```

$$-\frac{\hbar^2 \left(2\,\psi^{(1,0,0)}[r,\theta,\phi] + r\,\psi^{(2,0,0)}[r,\theta,\phi]\right)}{r}$$

```
Pr[r ψ[r, θ, ϕ]] - r Pr[ψ[r, θ, ϕ]] // Simplify
```

$-i\,\hbar\,\psi[r,\theta,\phi]$

```
Pr[r^2 ψ[r, θ, ϕ]] - r^2 Pr[ψ[r, θ, ϕ]] // Simplify
```

$-2\,i\,r\,\hbar\,\psi[r,\theta,\phi]$

```
Pr[r^n ψ[r, θ, ϕ]] - r^n Pr[ψ[r, θ, ϕ]] // Simplify
```

$-i\,n\,r^{-1+n}\,\hbar\,\psi[r,\theta,\phi]$

```
r Nest[Pr, ψ[r, θ, ϕ], 2] - Nest[Pr, r ψ[r, θ, ϕ], 2] -
  2 ⅈ ℏ Pr[ψ[r, θ, ϕ]] // Simplify
```

0

```
r Nest[Pr, ψ[r, θ, ϕ], 3] - Nest[Pr, r ψ[r, θ, ϕ], 3] -
  3 ⅈ ℏ Nest[Pr, ψ[r, θ, ϕ], 2] // Simplify
```

0

```
Pr[1/r ψ[r, θ, ϕ]] - 1/r Pr[ψ[r, θ, ϕ]] // Simplify
```

$$\frac{i \hbar \psi[r, \theta, \phi]}{r^2}$$

```
Pr[1/r^2 ψ[r, θ, ϕ]] - 1/r^2 Pr[ψ[r, θ, ϕ]] // Simplify
```

$$\frac{2 i \hbar \psi[r, \theta, \phi]}{r^3}$$

```
Nest[Pr, 1/r^2 ψ[r, θ, ϕ], 2] - 1/r^2 Nest[Pr, ψ[r, θ, ϕ], 2] //
 Simplify
```

$$-\frac{2 \hbar^2 \left(\psi[r, \theta, \phi] - 2 r \psi^{(1,0,0)}[r, \theta, \phi]\right)}{r^4}$$

```
Nest[Pr, 1/r ψ[r, θ, ϕ], 2] - 1/r Nest[Pr, ψ[r, θ, ϕ], 2] //
 Simplify
```

$$\frac{2 \hbar^2 \psi^{(1,0,0)}[r, \theta, \phi]}{r^2}$$

10. Example (Townsend)

We show that

$$\mathbf{L}^2 = \mathbf{r}^2\mathbf{p}^2 - (\mathbf{r}\cdot\mathbf{p})^2 + i\hbar(\mathbf{r}\cdot\mathbf{p})$$

((Proof))

$$\mathbf{L}^2 = -\hbar^2\left(\frac{1}{\sin^2\theta}\frac{\partial^2}{\partial^2\phi} + \cot\theta\frac{\partial}{\partial\theta} + \frac{\partial^2}{\partial\theta^2}\right)$$

$$\mathbf{r}^2\mathbf{p}^2 = -\hbar^2\left(\frac{1}{\sin^2\theta}\frac{\partial^2}{\partial\phi^2} + \cot\theta\frac{\partial}{\partial\theta} + \frac{\partial^2}{\partial\theta^2} + 2r\frac{\partial}{\partial r} + r^2\frac{\partial^2}{\partial r^2}\right)$$

$$-(\mathbf{r}\cdot\mathbf{p})(\mathbf{r}\cdot\mathbf{p}) = r\hbar^2(\frac{\partial}{\partial r} + r\frac{\partial^2}{\partial r^2})$$

$$i\hbar(\mathbf{r}\cdot\mathbf{p}) = \hbar^2 r\frac{\partial}{\partial r}$$

Thus, we have

$$\begin{aligned}
\mathbf{r}^2\mathbf{p}^2 - (\mathbf{r}\cdot\mathbf{p})^2 + i\hbar(\mathbf{r}\cdot\mathbf{p}) &= -\hbar^2(\frac{1}{\sin^2\theta}\frac{\partial^2}{\partial\phi^2} + \cot\theta\frac{\partial}{\partial\theta} + \frac{\partial^2}{\partial\theta^2} + 2r\frac{\partial}{\partial r} + r^2\frac{\partial^2}{\partial r^2}) \\
&\quad + \hbar^2(2r\frac{\partial}{\partial r} + r^r\frac{\partial^2}{\partial r^2}) \\
&= -\hbar^2(\frac{1}{\sin^2\theta}\frac{\partial^2}{\partial\phi^2} + \cot\theta\frac{\partial}{\partial\theta} + \frac{\partial^2}{\partial\theta^2} +) \\
&= \mathbf{L}^2
\end{aligned}$$

((**Mathematica**))

```
eq0 = Leq[ψ[r, θ, ϕ]] // FullSimplify
```

$$-\hbar^2\left(\mathrm{Csc}[\theta]^2\,\psi^{(0,0,2)}[r,\theta,\phi] + \mathrm{Cot}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi] + \psi^{(0,2,0)}[r,\theta,\phi]\right)$$

```
eq1 = r^2 Peq[ψ[r, θ, ϕ]] // Simplify
```

$$-\hbar^2\left(\mathrm{Csc}[\theta]^2\,\psi^{(0,0,2)}[r,\theta,\phi] + \mathrm{Cot}[\theta]\,\psi^{(0,1,0)}[r,\theta,\phi] + \psi^{(0,2,0)}[r,\theta,\phi] + 2\,r\,\psi^{(1,0,0)}[r,\theta,\phi] + r^2\,\psi^{(2,0,0)}[r,\theta,\phi]\right)$$

```
eq2 = -(r ur).p[(r ur).p[ψ[r, θ, ϕ]]] // Simplify
```

$$r\,\hbar^2\left(\psi^{(1,0,0)}[r,\theta,\phi] + r\,\psi^{(2,0,0)}[r,\theta,\phi]\right)$$

```
eq3 = ⅈ ℏ (r ur).p[ψ[r, θ, ϕ]] // Simplify
r ℏ^2 ψ^(1,0,0)[r, θ, ϕ]

eq123 = eq1 + eq2 + eq3 // Simplify
-ℏ^2 (Csc[θ]^2 ψ^(0,0,2)[r, θ, ϕ] +
    Cot[θ] ψ^(0,1,0)[r, θ, ϕ] + ψ^(0,2,0)[r, θ, ϕ])

eq0 - eq123 // Simplify
0
```

11. Runge-Lenz operator in spherical coordinates (Sakurai and Napolitano)

The detail of the Runge-Lenz method will be discussed elsewhere. Here we show that the commutation relations associated with the Runge-Lenz operator:

$$\mathbf{M} = \frac{1}{2\mu}(\mathbf{p}\times\mathbf{L} - \mathbf{L}\times\mathbf{p}) - \frac{Ze^2}{r}\mathbf{r}\,.$$

can be proved using the differential operators in the spherical coordinates (just one of the exercises). We note that

$$M_x = \frac{1}{2\mu}(p_y L_z - p_z L_y - L_y p_z + L_z p_y) - \frac{Ze^2}{r}x\,.$$

The Hamiltonian is defined by

$$H = \frac{\mathbf{p}^2}{2\mu} - \frac{Ze^2}{r}\,.$$

where μ is the reduced mass. We show that

(a)

$$[\mathbf{M}, H] = 0\,,$$

(b)

$$\mathbf{L}\cdot\mathbf{M} = 0 = \mathbf{M}\cdot\mathbf{L}\,,$$

(c)

$$\mathbf{M}^2 = \frac{2}{\mu}H(\mathbf{L}^2 + \hbar^2) + Z^2 e^4$$

((**Mathematica**))

```
Clear["Global`"];
ux = {Sin[θ] Cos[ϕ], Cos[θ] Cos[ϕ], -Sin[ϕ]};
uy = {Sin[θ] Sin[ϕ], Cos[θ] Sin[ϕ], Cos[ϕ]};
uz = {Cos[θ], -Sin[θ], 0};
ur = {1, 0, 0};
x = r ur.ux;
y = r ur.uy;
z = r ur.uz;
 Lap := Laplacian[#, { r, θ, ϕ}, "Spherical"] &;
Gra := Grad[#, { r, θ, ϕ}, "Spherical"] &;
Diva := Div[#, {r, θ,  ϕ}, "Spherical"] &;
Curla := Curl[#, {r, θ,  ϕ}, "Spherical"] &;
L := (-ⅈ ℏ ( Cross[(ur r), Gra[#]]) &) // Simplify;
Lx := (ux.L[#] &) // Simplify;
Ly := (uy.L[#] &) // Simplify;
Lz := (uz.L[#] &) // Simplify;
LP := (Lx[#] + ⅈ Ly[#]) & // Simplify;
LM = (Lx[#] - ⅈ Ly[#]) & // Simplify;
Leq := (Nest[Lx, #, 2] + Nest[Ly, #, 2] + Nest[Lz, #, 2]) &;
```

```
p := (-ⅈ ℏ Gra[#]) &;
px := (ux.p[#]) &;
py := (uy.p[#]) &;
pz := (uz.p[#]) &;
Peq := (Nest[px, #, 2] + Nest[py, #, 2] + Nest[pz, #, 2]) &;
prq := ((-ⅈ ℏ)/2 ur .Gra[#] + (-ⅈ ℏ)/2 Diva[# ur]) &;
pq := ℏ/(ⅈ r) D[r #, r] &;
```

$Mx = \frac{1}{2\mu} (pyLz - pzLy - Lypz + Lzpy) - \frac{Z e^2}{r} x$

$H = \frac{1}{2\mu} p^2 - \frac{Z e^2}{r}$

```
Mx := (1/(2 μ) (py[Lz[#]] - pz[Ly[#]]
        - Ly[pz[#]] + Lz[py[#]]) - (Z e^2)/r x #) & // FullSimplify;
```

```
My := (1/(2 μ) (pz[Lx[#]] - px[Lz[#]]
        - Lz[px[#]] + Lx[pz[#]]) - (Z e^2)/r y #) & // FullSimplify;

Mz := (1/(2 μ) (px[Ly[#]] - py[Lx[#]]
        - Lx[py[#]] + Ly[px[#]]) - (Z e^2)/r z #) & // FullSimplify;

H1 = (1/(2 μ) Peq[#] - (Z e^2)/r #) &;
```

$$My := \left(\frac{1}{2\mu}\left(pz[Lx[\#]] - px[Lz[\#]] - Lz[px[\#]] + Lx[pz[\#]]\right) - \frac{Z e^2}{r}\, y\, \#\right) \&\ //\ \text{FullSimplify};$$

$$Mz := \left(\frac{1}{2\mu}\left(px[Ly[\#]] - py[Lx[\#]] - Lx[py[\#]] + Ly[px[\#]]\right) - \frac{Z e^2}{r}\, z\, \#\right) \&\ //\ \text{FullSimplify};$$

$$H1 = \left(\frac{1}{2\mu} Peq[\#] - \frac{Z e^2}{r}\, \#\right) \&;$$

Proof of

[Mx,H]=0, [My,H]=0, [Mz,H]=0

```
Mx[H1[ψ[r, θ, ϕ]]] - H1[Mx[ψ[r, θ, ϕ]]] // FullSimplify
0
My[H1[ψ[r, θ, ϕ]]] - H1[My[ψ[r, θ, ϕ]]] // FullSimplify
0
Mz[H1[ψ[r, θ, ϕ]]] - H1[Mz[ψ[r, θ, ϕ]]] // FullSimplify
0
```

$M^2 = Mx^2 + My^2 + Mz^2$

$M^2 = \frac{2}{\mu}(H\,L^2 + \hbar^2 H) + Z^2 e^4$

```
eq1 = Mx[Mx[ψ[r, θ, ϕ]]] + My[My[ψ[r, θ, ϕ]]] +
    Mz[Mz[ψ[r, θ, ϕ]]] // FullSimplify;
s11 = ħ^2 H1[ψ[r, θ, ϕ]] // FullSimplify;
s12 = H1[Leq[ψ[r, θ, ϕ]]] // FullSimplify;
s13 = 2/μ (s11 + s12) + Z^2 e^4 ψ[r, θ, ϕ];
eq1 - s13 // FullSimplify
0
```

LxMx+LyMy+LzMz=0

```
Lx[Mx[ψ[r, θ, ϕ]]] + Ly[My[ψ[r, θ, ϕ]]] + Lz[Mz[ψ[r, θ, ϕ]]] //
 FullSimplify
```

```
0
```

MxLx + MyLy + MzLz = 0

```
0
```

```
Mx[Lx[ψ[r, θ, ϕ]]] + My[Ly[ψ[r, θ, ϕ]]] + Mz[Lz[ψ[r, θ, ϕ]]] //
 FullSimplify
```

```
0
```

MxLy - LyMx - iħMz=0

```
Mx[Ly[ψ[r, θ, ϕ]]] - Ly[Mx[ψ[r, θ, ϕ]]] - i ħ Mz[ψ[r, θ, ϕ]] //
 Simplify
```

```
0
```

$MxMy - MyMx + \frac{i2\hbar}{\mu} H\,Lz=0$

```
Mx[My[ψ[r, θ, ϕ]]] - My[Mx[ψ[r, θ, ϕ]]] +
  i (2 ħ/μ) H1[Lz[ψ[r, θ, ϕ]]] // Simplify
```

```
0
```

12. Conclusion

Using the above formula, one can derive the Schrödinger equation of hydrogen in the spherical coordinates. We will discuss this problem later.

REFERENCES

G.B. Arfken, H.J. Weber, and F.E. Harris, Mathematical Methods for Physicists, 7th edition (Academic Press, 2013).

A. Goswami, Quantum Mechanics, 2nd edition (Wm.C. Brown, 1997).

J.S. Townsend, A Modern Approach to Quantum Mechanics, 2nd edition (University Science Book, 2012).
M.S. Rogalski and S.B. Palmer, Quantum Physics (Gordon and Breach Science, 1999).